Microprocessors, Lecture 4

Z80 Assembly Language

Hamid Fadishei
Assistant Professor, University of Bojnord
Spring 2015

Assembly source structure



Some definitions

- Assembler
  - Converts the assembly program source code to executable machine code
    - Cross-assembler
      - Runs on a system other than the one you are programming for
    - Resident assembler (self-assembler)
      - Runs on a computer for which it assembles programs
- Two-pass assembler
  - Goes through the source twice
  - Sometimes only one pass is not enough to assemble
  - Example: Labels which are used before definition
- Loader
  - Takes the output of assembler (object code) and puts it into volatile memory (e.g. into SRAM)
- Programmer
  - Takes the output of assembler (hex code) and writes it into permanent memory (e.g. into FLASH)
- Pseudo operations
  - Do not directly translate to a machine instruction, but help the assembler to do its work
  - ORG, DB, DW, EQU, ...

Assembly programming



Assembly programming

```
        XOR     A               ; Clear A
        LD      B, 10           ; Set counter to 10
LOOP:   ADD     A, B            ; Add next number
        DEC     B               ; Prepare next number
        JP      NZ, LOOP        ; Jump if number is not zero
        HALT
```



Assembly programming

```
0001    0000 AF                 XOR     A               ; Clear A
0002    0001 06 0A              LD      B, 10           ; Set counter to 10
0003    0003 80         LOOP:   ADD     A, B            ; Add next number
0004    0004 05                 DEC     B               ; Prepare next number
0005    0005 C2 03 00           JP      NZ, LOOP        ; Jump if number is not zero
0006    0008 76                 HALT
```



Assembly programming



Assembly programming



Do we have to learn assembly?

- Today microprocessors can be programmed high-level
  - In old days, hand-writing assembly code usually gave more efficient programs
  - But today high-level compilers are very intelligent
  - Learning low-level programming concepts is still necessary
    - A base without which we may not fully understand concepts
    - You may need to reverse-engineer a machine code
    - You might not find a high level compiler for your microprocessor

Programming patterns: Basics

- Example: 1's complement
  - Logically complement the contents of memory location 0040H and place the result into memory location 0041H
- Solution

```
0001        LD      A, (0040H)      ; Get data
0002        CPL                     ; Complement
0003        LD      (0041H), A      ; Store result
0004        HALT
```

Programming patterns: Basics

- Example: 8-bit addition
  - Add the contents of memory location 0040H and 0041H, and place the result into memory location 0042H
- Solution 1

```
0001        LD      A, (0040H)      ; Get first operand
0002        LD      B, A            ; Save first operand
0003        LD      A, (0041H)      ; Get second operand
0004        ADD     A, B            ; Add operands
0005        LD      (42H), A        ; Store sum
0006        HALT
```

- Solution 2

```
0001        LD      HL, 0040H
0002        LD      A, (HL)         ; Get first operand
0003        INC     HL
0004        ADD     A, (HL)         ; Add second operand
0005        INC     HL
0006        LD      (HL), A         ; Store result
0007        HALT
```

Programming patterns: Basics

- Example: max(m,n)
  - Place the larger of the contents of memory locations 0040H and 0041H into memory location 0042H.
- Solution

```
0001          LD      HL, 0040H
0002          LD      A, (HL)          ; Get first operand
0003          INC     HL
0004          CP      (HL)             ; Is second operand larger?
0005          JR      C, CONT
0006          LD      A, (HL)          ; Yes, get second operand instead
0007 CONT:    INC     HL
0008          LD      (HL), A
0009          HALT
```

Programming patterns: Basics

- Example: 16-bit addition
  - Add the 16-bit number in memory location 0040H and 0041H to the number in memory location 0042H and 0043H. Store the result in 0044H and 0045H.
- Solution

```
0001        LD      HL, (0040H)     ; Get firest operand
0002        LD      DE, (0042H)     ; Get second operand
0003        ADD     HL, DE          ; Add numbers
0004        LD      (44H), HL       ; Store the result
0005        HALT
```

0040	13
	A1
0042	22
	17
0043	

0040	13
	A1
0042	22
	17
0043	35
	B8

A113 + 1722 = B835

Programming patterns: Table lookup

- Example: table lookup (square)
  - Calculate the square of the contents of memory location 0040H from a table and place it into location 0041H. The operand is between 0 and 7
- Solution

```
0001          LD      A, (0040H)      ; Get operand
0002          LD      L, A            ; Make data into 16-bit index
0003          LD      H, 0
0004          LD      DE, SQTAB       ; Get start address of table
0005          ADD     HL, DE          ; Index table with data
0006          LD      A, (HL)
0007          LD      (0041H), A
0008          HALT
0009
0010          ORG     50H     ; Square table
0011 SQTAB:   DEFB    0
0012          DEFB    1
0013          DEFB    4
0014          DEFB    9
0015          DEFB    16
0016          DEFB    25
0017          DEFB    36
0018          DEFB    49
```

Programming patterns: Loops

- Example: sum loop
  - Sum a series of numbers. The length of the series is in location 0041H and the series itself begins in location 0042H. The result should be stored in 0040H
- Solution

```
0001          LD      HL, 0041H
0002          LD      B, (HL)          ; count = length of series
0003          SUB     A                ; sum = 0
0004 SUMD:    INC     HL
0005          ADD     A, (HL)          ; sum = sum + data
0006          DEC     B
0007          JR      NZ, SUMD
0008          LD      (40H), A         ; Store sum
0009          HALT
```



- DJNZ
  - commonly used instruction in loops
  - Let's refer to handbook…

Programming patterns: Subroutines

- Example: MULT10 subroutine
  - Write a subroutine that multiplies by 10 a 16-bit number stored in HL
- Solution

```
0001 MULT10: PUSH    BC
0002         ADD     HL, HL
0003         LD      B, H
0004         LD      C, L
0005         ADD     HL, HL
0006         ADD     HL, HL
0007         ADD     HL, BC
0008         POP     BC
0009         RET
```

Programming patterns: Fromat conversion

- Purpose: Subroutine to convert BCD to binary
  - The (multi-byte) number to be converted is located in a memory location addressed by DE. The number is terminated by a non-numeric byte. The result should be returned in HL
- Solution

```
0001          ORG     200H
0002 BCDBIN:  LD      HL, 0           ; clear result
0003 LOOP:    LD      A, (DE)         ; next BCD byte
0004          CP      0AH
0005          JR      NC, FINISH      ; end of BCD number
0006          CALL    MULT10          ; HL = HL*10
0007          ADD     A, L            ; HL = HL + A
0008          LD      L, A
0009          LD      A, H
0010          ADC     A, 0
0011          LD      H, A
0012          INC     DE              ; Increment pointer
0013          JR      LOOP
0014 FINISH:  RET
```